Есть форма, на ней поле Согл, тип СправочникСсылка.СоглашенияСКлиентами
Есть Макет1 на СКД, сзапросом и выводом наименования в ТабДок:

```1c
    ВЫБРАТЬ
    СоглашенияСКлиентами.Наименование КАК Наименование
    ИЗ
    Справочник.СоглашенияСКлиентами КАК СоглашенияСКлиентами
    ГДЕ
    СоглашенияСКлиентами.Ссылка = &СсылкаНаСогл
```

Вывод данных на форму осуществляется процедурой:

```1c
    &НаСервере
    Процедура Тест1(ТабДок, сСогл)

    ТабДок.Очистить();
    ОтчетОбъект = РеквизитФормыВЗначение("Объект");

    СхемаКомпоновкиДанных = ОтчетОбъект.ПолучитьМакет("Макет1");

    СхемаКомпоновкиДанных.Параметры.Ссылка.Значение = сСогл;

    ДанныеРасшифровки = Новый ДанныеРасшифровкиКомпоновкиДанных;

    КомпоновщикМакета = Новый КомпоновщикМакетаКомпоновкиДанных;
    МакетКомпоновки = КомпоновщикМакета.Выполнить(СхемаКомпоновкиДанных,СхемаКомпоновкиДанных.НастройкиПоУмолчанию);

    ПроцессорКомпоновкиДанных = Новый ПроцессорКомпоновкиДанных;
    ПроцессорКомпоновкиДанных.Инициализировать(МакетКомпоновки,,ДанныеРасшифровки, Истина);

    ПроцессорВывода = Новый ПроцессорВыводаРезультатаКомпоновкиДанныхВТабличныйДокумент;
    ПроцессорВывода.УстановитьДокумент(ТабДок);

    ПроцессорВывода.Вывести(ПроцессорКомпоновкиДанных);

    КонецПроцедуры
```

Все отлично работает.
Сделала новый макет, поменяла запрос на более сложный:

```1c
    ВЫБРАТЬ
    ТоварыНаСкладахОстатки.Номенклатура КАК Номенклатура,
    ТоварыНаСкладахОстатки.ВНаличииОстаток КАК ВНаличииОстаток,
    Согл_ЦенГруппы.ВидЦен КАК Цена_СоглЦенГр,
    Согл_Товары.ВидЦены КАК Цена_СоглТовар,
    ЕСТЬNULL(СоглашенияСКлиентами.ВидЦен, &ВидЦенПоУмолчанию) КАК Цена_СоглОбщая,
    ЕСТЬNULL(Согл_Товары.ВидЦены, Согл_ЦенГруппы.ВидЦен) КАК Цена_Инд,
    ПОДСТРОКА(ВЫРАЗИТЬ(СсылкиНаФото.Значение КАК СТРОКА), 2, 50) КАК НашаСсылка
    ИЗ
    РегистрНакопления.ТоварыНаСкладах.Остатки(&Период, ) КАК ТоварыНаСкладахОстатки
    ЛЕВОЕ СОЕДИНЕНИЕ Справочник.СоглашенияСКлиентами.ЦеновыеГруппы КАК Согл_ЦенГруппы
    ПО ТоварыНаСкладахОстатки.Номенклатура.ЦеноваяГруппа = Согл_ЦенГруппы.ЦеноваяГруппа
    И (Согл_ЦенГруппы.Ссылка.Ссылка = &СсылкаНаСогл)
    ЛЕВОЕ СОЕДИНЕНИЕ Справочник.СоглашенияСКлиентами.Товары КАК Согл_Товары
    ПО ТоварыНаСкладахОстатки.Номенклатура = Согл_Товары.Номенклатура
    И (Согл_Товары.Ссылка.Ссылка = &СсылкаНаСогл)
    ЛЕВОЕ СОЕДИНЕНИЕ Справочник.СоглашенияСКлиентами КАК СоглашенияСКлиентами
    ПО (&СсылкаНаСогл = СоглашенияСКлиентами.Ссылка)
    И ТоварыНаСкладахОстатки.Склад = СоглашенияСКлиентами.Склад
    ЛЕВОЕ СОЕДИНЕНИЕ Справочник.Номенклатура.ДополнительныеРеквизиты КАК СсылкиНаФото
    ПО ТоварыНаСкладахОстатки.Номенклатура = СсылкиНаФото.Ссылка.Ссылка
    И (&СвФото = СсылкиНаФото.Свойство)
```

Проверила в консоли - запрос работает и возвращает данные.
Сделала новую процедуру:

```1c
    &НаСервере
    Процедура СформироватьМакет2(ТабДок, сСогл)

    ТабДок.Очистить();
    ОтчетОбъект = РеквизитФормыВЗначение("Объект");

    СхемаКомпоновкиДанных = ОтчетОбъект.ПолучитьМакет("Макет2");

    СхемаКомпоновкиДанных.Параметры.СсылкаНаСогл.Значение = сСогл;
    СхемаКомпоновкиДанных.Параметры.Период.Значение = ТекущаяДата();
    СхемаКомпоновкиДанных.Параметры.ВидЦенПоУмолчанию.Значение = Справочники.ВидыЦен.НайтиПоНаименованию("Прайс", истина);
    СхемаКомпоновкиДанных.Параметры.СвФото.Значение = ПланыВидовХарактеристик.ДополнительныеРеквизитыИСведения.НайтиПоНаименованию("Фото", истина);

    ДанныеРасшифровки = Новый ДанныеРасшифровкиКомпоновкиДанных;

    КомпоновщикМакета = Новый КомпоновщикМакетаКомпоновкиДанных;
    МакетКомпоновки = КомпоновщикМакета.Выполнить(СхемаКомпоновкиДанных,СхемаКомпоновкиДанных.НастройкиПоУмолчанию);

    ПроцессорКомпоновкиДанных = Новый ПроцессорКомпоновкиДанных;
    ПроцессорКомпоновкиДанных.Инициализировать(МакетКомпоновки,,ДанныеРасшифровки, Истина);

    ПроцессорВывода = Новый ПроцессорВыводаРезультатаКомпоновкиДанныхВТабличныйДокумент;
    ПроцессорВывода.УстановитьДокумент(ТабДок);

    ПроцессорВывода.Вывести(ПроцессорКомпоновкиДанных);

    КонецПроцедуры
```

Получаю ошибку:

В чем может быть причина? Запрос отрабатывается в консоли "на ура"
---
Смотрите ваш СКД, закладка: Параметры
...проверьте ваш запрос в консоли
---
а зачем так сложно? сделайте формирование вашего ТабДока как формирование обычной печатной формы!!! там все просто и понятно
---
Там условное форматирование уже удобно настроено через СКД. Я изначально сделала обычный отчет с форматированием на СКД, все работает, красиво выводится. Осталось передать параметры - и на тебе...
---
по поводу ошибки,
так вам же 1С сообщила что не задан параметр: СсылкаНаСогл

по причине:
{(14, 38)}: Не задано значение параметра "СсылкаНаСогл"
---
Смотрите ваш СКД, закладка: Параметры
...проверьте ваш запрос в консоли
---
Спасибо, надежда была только на закладку Параметры.
Поменяла Использование - авто на Использование - всегда
Заработало, спасибо!
---
ПО ТоварыНаСкладахОстатки.Номенклатура = СсылкиНаФото.Ссылка.Ссылка

А почему не СсылкиНаФото.Ссылка.Ссылка.Ссылка.Ссылка?

В смысле, не надо так делать. СсылкиНаФото.Ссылка достаточно. Дальше уже идет еще одно неявное присоединение таблицы СсылкиНаФото, оно нам надо?